
1 
 

Министерство образования и науки Российской Федерации 
ФГБОУ ВПО "Ивановский государственный университет" 

Факультет математики и компьютерных наук 
Кафедра алгебры и математической логики 

 
 
 

 
 

АЛГЕБРАИЧЕСКИЕ ВЫЧИСЛЕНИЯ 
В СИСТЕМЕ SAGE 

 
 
 

Методические указания по дисциплинам  
"Фундаментальная алгебра" и "Компьютерная алгебра" 

для студентов 2 курса 
факультета математики и компьютерных наук 

(квалификация "Бакалавр") 
 
 
 
 
 
 
 
 
 
 
 

Иваново 
Издательство "Ивановский государственный университет" 

2014 
  



2 
 

Составитель: кандидат физико-математических наук, профессор 
 кафедры алгебры и математической логики Н. И. Яцкин 
Под редакцией заведующего кафедрой прикладной математики  
 и компьютерных наук Е. В. Соколова 
 
 В пособии демонстрируются возможности применения одной из со-
временных систем компьютерной алгебры, SAGE, при выполнении типовых 
вычислений с такими объектами абстрактной алгебры как группы, кольца, 
поля. Издание допускает использование ознакомительного характера в про-
цессе освоения студентами соответствующего теоретического материала в 
курсе "Фундаментальная алгебра", а также, в следующем семестре, – ак-
тивное практическое использование при изучении алгебраических алгорит-
мов (и их программных реализаций) в курсе "Компьютерная алгебра". 
 
 
 
 
 

Печатается по решению методической комиссии 
факультета математики и компьютерных наук 

Ивановского государственного университета 
 
 

Рецензент 
кандидат физико-математических наук С. И. Хашин (ИвГУ) 

 
 
 
 
 
 
 

 Яцкин Н. И. 
составление, 2014 

 ФГБОУ ВПО "Ивановский  
государственный университет", 

2014 
  



3 
 

 
 

Содержание 
 

Ведение 4 
1. Элементы линейной алгебры 6 

1.1. Матрицы, векторные пространства и системы линейных 
уравнений 

 
6 

1.2. Функции от матриц. Преобразования и канонические 
формы матриц  

 
11 

1.3. Матрицы над евклидовыми кольцами. Каноническая 
форма Смита 

 
16 

2. Элементы общей (абстрактной) алгебры 20 
2.1. Поля, кольца, идеалы в кольцах, фактор-кольца 20 
2.2. Кольца многочленов и фактор-кольца для них 25 
2.3. Конечные поля 30 
2.4. Группы 36 

Интернет-ресурсы и литература 45 
 
 

  



4 
 

 
Введение 
 

 Предлагаемое учебное пособие призвано помочь студентам, изучаю-
щим учебную дисциплину "Компьютерная алгебра", в практическом освое-
нии важнейших приемов оперирования с абстрактными алгебраическими 
объектами (такими как группы, кольца, поля), в рамках специально скон-
струированной для этих целей, современной и постоянно совершенствуемой 
(усилиями широких слоев математического сообщества), свободно распро-
страняемой системы компьютерной алгебры Sage. На этой базе предполага-
ется дальнейшее изучение основных точно работающих алгебраических ал-
горитмов.  
 Предварительное знакомство с идеологией и языком Sage можно осу-
ществить с помощью методического пособия [8], подготовленного студента-
ми факультета математики и компьютерных наук ИвГУ А. Е. Куваевым и 
А. С. Смоляковым, которое, в частности, может служить детальным руковод-
ством по установке системы и, кроме того, содержит первоначальные сведе-
ния об основах Sage-программирования. (Именно благодаря этой книжке 
автору настоящего пособия удалось установить Sage на своем компьютере и 
написать первую Sage-процедуру.)  
 Автором использовалась новейшая (на момент написания пособия) 
версия 6.0 системы; полный комплект документации доступен на главной 
странице проекта Sage (см. [1, 2, 6]). Весьма интересные методики исполь-
зования Sage при изучении стандартных университетских курсов алгебры 
представлены в учебных пособиях [3 – 5, 7]. Укажем также те учебники по 
чисто алгебраическим курсам, к которым были "привязаны" позднейшие 
Sage-разработки: [9 - 11]. 
 Вообще, предполагается, что определения и свойства основных алгеб-
раических структур (объектов), упоминаемых в данном пособии, либо уже 
известны читателям (например, из курса фундаментальной алгебры), либо 
могут быть оперативно подгружены в их оперативную память (из достаточно 
богатой учебной литературы или из сети; см., в частности, русскоязычные 
издания [12 - 22]). Некоторые, наиболее важные, аспекты теории кратко 
напоминаются и поясняются, однако более полную и строгую теоретическую 
информацию следует искать в указанных источниках. 
 Автор, не будучи по образованию специалистом в компьютерных 
науках, но заинтересованный, тем не менее, во внедрении, современных (ос-
нованных на использовании систем компьютерной алгебры) образователь-
ных технологий в учебный процесс (в том числе - и по базовым математиче-

ским дисциплинам), полагает важным упомянуть, что весьма полезными в 
ходе подготовки настоящего пособия были обсуждения предварительных ва-
риантов текста с Е. В. Соколовым и другими коллегами, а также - с наиболее 
"продвинутыми" студентами-компьютерщиками, и, в частности, - с соавто-
рами пособия [8].  
 Возможно, читателям будет полезно сразу увидеть 

1. Элементы линейной алгебры  
 

1.1. Матрицы, векторные пространства и системы  
 линейных уравнений 

 

Предусмотрено несколько способов ввода матриц. Представим про-
стейший и самый естественный из них: матрица задается построчно, как спи-
сок списков. Обратите внимание на то, что в качестве первого аргумента фи-
гурирует имя поля (или кольца), над которым рассматривается матрица. 
Sage узнает по имени несколько важнейших колец и полей:  

ZZ = ℤ, кольцо целых чисел; 
QQ = ℚ, поле рациональных чисел; 
RR = ℝ, поле действительных чисел; 
CC = ℂ, поле комплексных чисел; 

другие примеры будут описаны далее, по мере необходимости. Сразу акцен-
тируем внимание на том, что в первых двух примерах кольцо и поле являют-
ся точными (в том смысле, что имеется принципиальная возможность абсо-
лютно точного представления для их элементов); в двух последующих при-
мерах поля являются приближенными (для записи их элементов приходится 
прибегать к аппроксимациям, с той или иной степенью точности). 
 

Пример 1.1.1. Решение системы линейных уравнений.  
 

sage: A=matrix(QQ,[[1,4,0,-1,0,7,-9], 
      [2,8,-1,3,9,-13,7], 
      [0,0,2,-3,-4,12,-8], 
      [-1,-4,2,4,8,-31,37]]); A;  
 

[   1   4   0  -1   0   7  -9] 
[   2   8  -1   3   9 -13   7] 
[   0   0   2  -3  -4  12  -8] 
[  -1  -4   2   4   8 -31  37] 
 

 Определим размеры и ранг введенной матрицы. 
 

sage: m,n=A.nrows(),A.ncols(); m,n; A.rank() 
 

(4, 7) 
3 
 

 Далее вводится вектор-столбец, "высота" которого совпадает с коли-
чеством строк ранее введенной матрицы. Sage записывает его в строчку, но 
не в квадратных (как положено для списков), а в круглых скобках.  
 

sage: b=vector(QQ,[3,9,1,4]); b 
 

(3, 9, 1, 4) 

 Рассмотрим теперь систему линейных уравнений (с.л.у.) ܣ ∙ ݔ̅ = തܾ, со-
держащую ݉ = 4 уравнений с ݊ = 7 неизвестными. Для ее исследования об-
разуем расширенную (аугментированную) матрицу системы ܤ = |ܣ) തܾ).  
 (Разграничивающая черта в матрице появляется благодаря применению 
опции subdivide=True; разумеется, она не обязательна, ни на что не вли-
яет и используется исключительно в обучающих примерах.)  
 

sage: B=A.augment(b,subdivide=True); B; B.rank() 
 

[   1   4   0  -1   0   7  -9|   3] 
[   2   8  -1   3   9 -13   7|   9] 
[   0   0   2  -3  -4  12  -8|   1] 
[  -1  -4   2   4   8 -31  37|   4] 
 

3 
 Вычислив ранг матрицы ܤ и пользуясь теоремой Кронекера-Капелли, 
мы можем заключить, что равенство rank(ܣ) = rank(ܤ) влечет совмест-
ность рассматриваемой с.л.у. Далее, поскольку количество уравнений мень-
ше количества неизвестных, эта система является неопределенной и, следова-
тельно, в ее общем решении должны присутствовать свободные неизвестные, 
в количестве, равном ݊ − rank(ܣ). 
 Переходим к практическому отысканию общего решения. Приведем 
матрицу ܤ к виду Жордана-Гаусса. В англоязычной математической литера-
туре этот вид принято называть "приведенной, эшелонированной по строкам 
формой" (rref = Reduced Row-Echelon Form). 
 

sage: B.rref() 
 

[  1  4  0  0  2  1 -3|  4] 
[  0  0  1  0  1 -3  5|  2] 
[  0  0  0  1  2 -6  6|  1] 
[  0  0  0  0  0  0  0|  0] 
 

 Визуально наблюдаем главные (ключевые) столбцы, содержащие (еди-
ничные) ключевые элементы. По-английски последние называются pivots 
(загляните в англо-русский словарь: pivot = штырь, опорный пункт, ключе-
вой игрок и т. п.; математический словарь добавляет: основной, разрешаю-
щий, опорный элемент). В Sage предусмотрены методы для вывода переч-
ней номеров ключевых и неключевых (свободных) столбцов. Количество 
ключевых столбцов – это ранг ݎ = rank(ܣ) = rank(ܤ), количество неключе-
вых столбцов (ݏ = ݊ − -равно количеству свободных неизвестных. (Не за (ݎ
бываем, что всё и всюду в Sage нумеруется, начиная с нуля.) 
 

sage: princ=B.pivots(); princ; r=len(princ); r 
 

(0, 2, 3) 
3 

sage: free=A.nonpivots(); free; s=len(free); s 
 

(1, 4, 5, 6) 
4 
 Sage умеет находить одно частное решение с.л.у., называемое опор-
ным, принцип получения которого таков: свободные неизвестные приравни-
ваются нулю. 
 

sage: z=A.solve_right(b); z 
 

(4, 0, 2, 1, 0, 0, 0) 
 

 Мы применили "правый" метод решения с.л.у.; есть еще "левый" метод, 
.solve_left, разрешающий с.л.у. ̅ݔ௧ ∙ ܣ = 0ത௧, в которой неизвестные запи-
сываются не в столбец, но в строку. 
 Если с.л.у. является однородной ( തܾ = 0ത), то будет выдано "неинтерес-
ное" (тривиальное, т. е. нулевое) решение. Для получения фундаментальной 
матрицы, содержащей базисные решения однородной с.л.у., применяется 
следующий метод вычисления так называемого ядра (нуль-пространства, 
kernel) матрицы. 
 

sage: K=A.right_kernel(basis='pivot'); K 
 

Vector space of degree 7 and dimension 4 over  
 Rational Field User basis matrix: 
 

[ -4  1  0  0  0  0  0] 
[ -2  0 -1 -2  1  0  0] 
[ -1  0  3  8  0  1  0] 
[  3  0 -5 -6  0  0  1] 
 

 Найден и записан (по строкам) в матрицу ܭ базис в (правом) ядре мат-
рицы ܣ. Sage описывает это ядро как подпространство размерности 4 в 
арифметическом линейном пространстве ℚ଻ размерности 7 (размерность 
объемлющего пространства Sage называет "степенью", degree). Далее вы-
численный выше базис нам понадобится представить как бы "россыпью", 
в виде последовательности векторов-столбцов. 
 

sage: N=K.basis(); N 
[ 
(-4, 1, 0, 0, 0, 0, 0), 
(-2, 0,-1,-2, 1, 0, 0), 
(-1, 0, 3, 6, 0, 1, 0), 
( 3, 0,-5,-6, 0, 0, 1) 
] 
 Как известно, о.р.н. (общее решение неоднородной с.л.у.) складывается 
из о.р.о. (общего решения соответствующей однородной с.л.у.) и произволь-
ного ч.р.н. (частного решения исходной неоднородной с.л.у.); ч.р.н. z уже 

найдено; о.р.о. представляется в виде линейной комбинации базисных част-
ных решений однородной системы (ч.р.о., которые также уже вычислены; 
см. последовательность N); в качестве коэффициентов линейной комбинации 
должны фигурировать переменные (свободные неизвестные, номера которых 
нам известны; см. список free). 
 По правилам Sage каждая переменная должна объявляться, поэтому 
сейчас мы определим кольцо QQx многочленов над полем QQ от семи пере-
менных (неизвестных в рассматриваемой с.л.у); сами эти переменные будут 
фигурировать как порождающие элементы (генераторы, gens) построенно-
го кольца. 
 

sage: QQx=PolynomialRing(QQ,n,'x'); QQx; QQx.gens() 
 

Multivariate Polynomial Ring in x0,x1,x2,x3,x4,x5,x6 
 over Rational Field 
(x0, x1, x2, x3, x4, x5, x6) 
 

 Теперь мы создадим (в цикле for) список, элементами которого будут 
векторы – найденные выше базисные ч.р.о., причем они будут рассматри-
ваться уже не над полем QQ, но над (более широким) кольцом многочленов 
QQx, хотя ни один из этих векторов пока не содержит ни одной из перемен-
ных. Переменные появятся на следующем шаге, как коэффициенты линейной 
комбинации. 
 

sage: f=[vector(QQx,n,N[i]) for i in range(m)]; f 
 

[(-4,1,0,0,0,0,0),(-2,0,-1,-2,1,0,0),(-1,0,3,6,0,1,0), 
 (3,0,-5,-6,0,0,1)] 
 

 Выберем из числа переменных те, которые являются свободными неиз-
вестными и вычислим сумму произведений свободных неизвестных на соот-
ветствующие базисные ч.р.о., плюс опорное ч.р.н. 
 

sage: y=[QQx.gen(free[i]) for i in range(m)]; y 
 

[x1, x4, x5, x6] 
 

sage: x=sum([y[i]*f[i] for i in range(m)])+z; x 
 

(-4*x1-2*x4-x5+3*x6+4, x1, -x4+3*x5-5*x6+2,  
 -2*x4+6*x5-6*x6+1, x4, x5, x6) 
 

 Наконец, о.р.н. x найдено; проверку можно выполнить, вычислив (над 
кольцом многочленов!) невязку.  
 

sage: A*x-b 
(0, 0, 0, 0) 
 Разбирая алгоритм решения с.л.у., мы уже сталкивались с необходимо-
стью рассмотрения линейных пространств (над различными полями). Так 

ядро (݉ × ݊)-матрицы ܣ является линейным подпространством, размерность 
которого равна ݊ − rank(ܣ), в арифметическом линейном пространстве ℚ௡. 
По матрице ܣ определяется еще одно линейное подпространство – образ этой 
матрицы, понимаемый как линейная оболочка (span) векторов-столбцов 
матрицы. Это подпространство (в примере оно обозначено W) располагается 
(вообще говоря) уже в другом линейном пространстве ℚ௠. Размерность об-
раза совпадает с рангом матрицы. 
 

Пример 1.1.2. Базис в подпространстве, порожденном столбцами матрицы. 
(Матрица A - из предыдущего примера). 
 

sage: V=QQ^4; V; W=V.span(A.columns()); W 
 

Vector space of dimension 4 over Rational Field 
Vector space of degree 4 and dimension 3 over  
 Rational Field Basis matrix: 
 

[  1  0  0 -31/7] 
[  0  1  0  12/7] 
[  0  0  1  13/7] 
 

sage: M=W.basis(); M; d=W.dimension(); d 
[ 
( 1, 0, 0,-31/7), 
( 0, 1, 0, 12/7), 
( 0, 0, 1, 13/7) 
] 
3 
 Матрицы фиксированного размера над заданным полем сами образуют 
векторное пространство над этим полем. Его можно явно задать и после это-
го выполнять в нем все допустимые в векторных пространствах операции. 
Ниже показано, например, как вызывается естественный базис в простран-
стве матриц заданного размера. 
 

Пример 1.1.3. Линейное пространство матриц заданного размера. 
 

sage: W=MatrixSpace(QQ,2,3); B=W.basis() 
 

Full MatrixSpace of 2 by 3 dense matrices over  
 Rational Field 
[ 
[1 0 0] [0 1 0] [0 0 1] [0 0 0] [0 0 0] [0 0 0] 
[0 0 0],[0 0 0],[0 0 0],[1 0 0],[0 1 0],[0 0 1] 
] 

1.2. Функции от матриц. Преобразования и  
 канонические формы матриц  

 

 В линейной алгебре изучаются разнообразные функции, за-
висящие от матричного аргумента. Некоторые из этих функций 
определены для произвольных (прямоугольных) матриц; некото-
рые – лишь для квадратных, причем не обязательно для всех. Некоторые 
функции принимают скалярные значения (ранг, определитель, след и т. п.); 
другие сопоставляют матрице новую матрицу (транспонированную, присо-
единенную, обратную и т. д.). Могут сопоставляться матрицам и более слож-
ные (составные) объекты: характеристический и минимальный многочлены, 
спектр (как множество ее собственных значений, возможно, - с учетом их 
кратностей, алгебраических и геометрических).  
 Многие из названных (и неназванных) функций легко реализуемы в 
компьютерных алгебраических системах (и, в том числе, в Sage); реализация 
других (над неточными полями) весьма трудоемка. 
 

Пример 1.2.1. Спектральные характеристики матрицы. 
 

 Начнем с того, что сгенерируем случайную (4 × 4)-матрицу ܣ с элемен-
тами из поля рациональных чисел [синтаксис соответствующей команды 
предусматривает ограничения на числители (num = numеrator) и знаме-
натели (den = denominator) матричных элементов] и применим к этой 
матрице упомянутые выше функции.  
 

sage: A=random_matrix(QQ,4,4,num_bound=4,den_bound=5);A 
 

[ -1 -3/2   -1    1] 
[  2    3    0   -3] 
[  0  4/5   -2    3] 
[ -2   -1  3/2  2/3] 
 

sage: A.rank(),A.det(),A.trace() 
 

(4, 314/15, 2/3) 
 

sage: A.transpose(),A.adjoint(),A.inverse() 
( 
[  -1   2    0   -2]  [-151/10 -689/60  -13/4  -72/5] 
[-3/2   3  4/5   -1]  [   -1/3    47/6   37/6      8] 
[  -1   0   -2  3/2]  [-236/15 -106/15     -2    4/5] 
[   1  -3    3  2/3], [  -52/5   -34/5      4   -8/5], 
 

[ -453/628 -689/1256 -195/1256  -108/157] 
[   -5/314   235/628   185/628    60/157] 
[ -118/157   -53/157   -15/157     6/157] 
[  -78/157   -51/157    30/157   -12/157] 
) 

 Отметим, что обратную матрицу можно было бы вычислить иначе: 
A^(-1). Чтобы найти характеристический многочлен ℎ஺(ݐ) = det(ܧݐ −  ,(ܣ
мы предварительно объявляем переменную t.  
 

sage: t=var('t'); h=A.charpoly(t); h; h.roots() 
 

t^4 - 2/3*t^3 - 19/2*t^2 + 163/15*t + 314/15 
[] 
 Попытка найти собственные значения (характеристические корни) 
приводит к пустому списку: в поле QQ их нет. Есть, однако, возможность 
расширить (алгебраически замкнуть) поле рациональных чисел.  
 Всякое поле ܲ можно вложить в некоторое алгебраически замкнутое 
поле. Наименьшее из таких полей определено однозначно (с точностью до 
изоморфизма); оно называется алгебраическим замыканием поля ܲ и обозна-
чается  തܲ.  
 Из начального курса алгебры известно, что алгебраическим замыкани-
ем поля действительных чисел является поле комплексных чисел: ℝഥ = ℂ. По-
ле рациональных чисел ℚ является подполем в ℝ, и поэтому его алгебраиче-
ское замыкание ८ = ℚഥ (именуемое полем алгебраических чисел) является 
подполем в ℂ. В более подробных курсах алгебры доказывается, что поле ८ 
получается присоединением к ℚ всех корней всевозможных многочленов с 
целочисленными коэффициентами.  
 В системе Sage для поля ८ используется обозначение QQbar; обозна-
чение AA резервируется для подполя ८ ∩ ℝ действительных алгебраических 
чисел. Переход от одного основного кольца (или поля) к другому осуществ-
ляется следующим образом. 
 

sage: A1=A.change_ring(QQbar); A1 
 

[  -1 -3/2   -1    1] 
[   2    3    0   -3] 
[   0  4/5   -2    3] 
[  -2   -1  3/2  2/3] 
 

 Матрица, разумеется, не изменилась, но теперь ее элементы рассматри-
ваются в более широком поле QQbar (которое система Sage относит к кате-
гории приближенных). 
 

sage: A1.charpoly(t).roots() 
 

[(-2.967556346854603?, 1),(-1.093191009111068?, 1), 
(2.363707011316169? - 0.9303855331002023?*I, 1), 
(2.363707011316169? + 0.9303855331002023?*I, 1)] 
 

 Мы получили четыре попарно различных (однократных) корня: два 
действительных и два комплексно сопряженных друг другу (мнимую едини-

цу Sage кодирует символом I); ответ выдан в виде списка пар; первый эле-
мент пары – это корень, второй – кратность корня.  
 В случае, когда квадратная матрица имеет полный набор попарно раз-
личных собственных значений, говорят, что она имеет простой спектр. Это 
условие является достаточным для диагонализируемости матрицы.  
 (Напомним, что квадратная матрица называется диагонализируемой, 
если она подобна некоторой диагональной матрице; отношение подобия 
ܣ ≈ ܤ определяется формулой ܤ = ܶିଵܶܣ, где ܶ - некоторая обратимая мат-
рица.)  
 В примере мы можем сделать вывод о диагонализируемости данной 
матрицы над полем алгебраических чисел (над полем рациональных чисел 
она не является диагонализируемой). Sage-тест может подтвердить данный 
вывод. 
 

sage: A1.is_diagonalizable(); A.is_diagonalizable() 
 

True 
Traceback (click to the left of this block for 
 traceback) 
... 
RuntimeError: an eigenvalue of the matrix is not 
 contained in Rational Field 
 

 Обратите внимание на то, что иногда, вместо естественного False 
(или, в других случаях, - вместо пустого ответа), Sage запрограммирован на 
выдачу сообщения об ошибке. 
  

 Рассмотрим со всеми подробностями задачу о приведении квадратной 
матрицы к жордановой нормальной форме (ж.н.ф.). 
 

Пример 1.2.2. Приведение матрицы к жордановой нормальной форме. 
 

sage: A=matrix(QQ,[[-9,-5,-3,0,-8,-3,-7,13], 
      [1,0,2,1,2,-1,2,-3], 
      [-6,-6,-5,1,-9,-4,-5,12], 
      [0,1,0,-1,1,-1,1,0], 
      [-1,-1,-1,0,-3,0,-1,2], 
      [1,1,1,1,1,-3,2,-2], 
      [1,0,1,0,0,0,-1,-2], 
      [-6,-5,-2,1,-8,-4,-5,9]]); A 
 

[-9 -5 -3  0 -8 -3 -7 13] 
[ 1  0  2  1  2 -1  2 -3] 
[-6 -6 -5  1 -9 -4 -5 12] 
[ 0  1  0 -1  1 -1  1  0] 
[-1 -1 -1  0 -3  0 -1  2] 
[ 1  1  1  1  1 -3  2 -2] 

[ 1  0  1  0  0  0 -1 -2] 
[-6 -5 -2  1 -8 -4 -5  9] 
 

 Вычислим и разложим на неприводимые множители (метод .factor) 
характеристический многочлен.  
 

sage: t=var('t'); h=A.charpoly(t); h; h.factor() 
 

t^8+13*t^7+70*t^6+196*t^5+280*t^4+112*t^3-224*t^2 
 -320*t-128 
(t-1)*(t+2)^7 
 

 Из разложения определяем собственное значение ݐଵ = 1 кратности 
݉ଵ = 1 и собственное значение ݐଶ = 1 кратности ݉ଶ = 7. Сумма (алгебраи-
ческих) кратностей собственных значений равна размерности пространства 
(размеру матрицы): ݉ଵ + ݉ଶ = 8. Это свидетельствует о том, что данная 
матрица приводима к жордановой нормальной форме (ж.н.ф.) над полем ра-
циональных чисел.  
 Предусмотрены специализированные методы: .eigenvalues - вы-
числяет собственные значения; .eigenspaces_right – описывает соот-
ветствующие собственные подпространства (как линейные оболочки; ука-
зываются базисные собственные векторы). Снова существуют две версии 
методов: левая и правая; наш выбор, привычный по курсу линейной алгеб-
ры, - правая версия. Кроме того, следует обратить внимание на уточнение 
'all': ищутся все собственные подпространства. 
 

sage: A.eigenvalues();  
 

[1,-2,-2,-2,-2,-2,-2,-2] 
 

sage: A.eigenspaces_right(format='all') 
 

[ 
(1, Vector space of degree 8 and dimension 1 over  
 Rational Field User basis matrix: 
 [1 0 1 0 0 0 0 1]), 
(-2, Vector space of degree 8 and dimension 3 over  
 Rational Field User basis matrix: 
 [ 1  0  0  1  0  0 -1  0] 
 [ 0  1  0  1 -1  1  0  0] 
 [ 0  0  1  0  0  1  1  1]) 
] 
 Теперь нам видны геометрические кратности собственных значений 
(т. е. размерности собственных подпространств): ݊ଵ = 1, ݊ଶ = 3; их сумма 
݊ଵ + ݊ଶ < 8 . Вспоминая критерий диагонализируемости матрицы (сумма 
геометрических кратностей собственных значения должна равняться размер-

ности пространства), заключаем, что данная матрица не является диагонали-
зируемой. (Можете спросить Sage, он скажет то же самое.) 
 Найдем ж.н.ф. ܬ матрицы ܣ, вместе с матрицей перехода ܶ, реализую-
щей подобие ܬ = ܶିଵܶܣ. 
 

sage: J,T=A.jordan_form(transformation=True); J,T 
 

( 
[ 1| 0  0  0  0| 0  0| 0]  
[--+-----------+-----+--]  
[ 0|-2  1  0  0| 0  0| 0]  
[ 0| 0 -2  1  0| 0  0| 0]  [1  1  2 0 0 -1/3    1 0] 
[ 0| 0  0 -2  1| 0  0| 0]  [0  0  1 1 0  1/3    0 0] 
[ 0| 0  0  0 -2| 0  0| 0]  [1  0  0 1 0    0    0 1] 
[--+-----------+-----+--]  [0  1  1 1 1    0    0 0] 
[ 0| 0  0  0  0|-2  1| 0]  [0  0  0 0 0 -1/3    0 0] 
[ 0| 0  0  0  0| 0 -2| 0]  [0  0  0 1 0  1/3 -2/3 1] 
[--+-----------+-----+--]  [0 -1 -1 0 0  1/3 -2/3 1] 
[ 0| 0  0  0  0| 0  0|-2], [1  0  1 1 0    0    0 1] 
) 
 

 Проверка. 
 

sage: (J-T^(-1)*A*T).is_zero() 
 

True 
 

 Sage старательно подразделяет матрицу J на блоки; жордановы ящи-
ки очень хорошо видны. Заметим, однако, что (даже и при не очень больших 
размерах матриц) стремление Sage сделать все столбцы одинаковыми по 
ширине приводит к хаотическому и трудно читаемому разбросу элементов по 
странице. В таких случаях лучше не выводить матрицы простым указанием 
их имен, но поступить чуть хитрее: в двойном цикле перебрать все элементы. 
(Полученный список списков при необходимости легко преобразуется об-
ратно в матрицу.) 
 

sage: [[T[i,j] for j in range(T.ncols())]  
               for i in range(T.nrows())]  
 

[[1,1,2,0,0,-1/3,1,0],[0,0,1,1,0,1/3,0,0], 
 [1,0,0,1,0,0,0,1],[0,1,1,1,1,0,0,0], 
 [0,0,0,0,0,-1/3,0,0],[0,0,0,1,0,1/3,-2/3,1], 
 [0,-1,-1,0,0,1/3,-2/3,1],[1,0,1,1,0,0,0,1]] 

1.3. Матрицы над евклидовыми кольцами. Каноническая  
 форма Смита  
 
 Для матриц над полем основным рабочим инструментом 
служит алгоритм Жордана-Гаусса приведения к виду Жордана-
Гаусса (rref, см. п. 1.1), основанный на применении к строкам матрицы так 
называемых элементарных преобразований трех типов: (1) перестановка 
строк; (2) прибавление к некоторой строке другой строки, домноженной на 
некоторый скаляр; (3) умножение некоторой строки на ненулевой скаляр. Ес-
ли разрешить аналогичные преобразования над столбцами, то любую прямо-
угольную матрицу можно привести к так называемому скелетному виду: 
в начале главной диагонали располагаются единицы, в количестве, равном 
рангу исходной матрицы, остальные элементы равны нулю. Кстати, вы мо-
жете легко, пользуясь справочником или "хелпами", получить информацию о 
Sage-методах, реализующих элементарные преобразования всех типов. 
 При работе с матрицами, элементы которых принадлежат не полю, но, 
скажем, коммутативному кольцу ܭ, возникает одно, но очень важное отли-
чие: в преобразованиях типа (3), над строками или столбцами, скаляр обязан 
быть обратимым элементом кольца ܭ (а в кольце обратимыми могут ока-
заться не все, а лишь некоторые из ненулевых элементов; они образуют 
группу, которая обозначается ܭ∗ и называется мультипликативной группой 
кольца ܭ). Наиболее простой вид соответствующая теория приобретает в 
случае так называемых евклидовых колец. Евклидовыми называются целост-
ные кольца, в которых может быть задан некоторый алгоритм деления с 
остатком, причем для ненулевых элементов кольца должна быть определена 
некоторая "следящая функция" (обычно именуемая евклидовой нормой) та-
кая, что остаток либо равен нулю, либо его норма строго меньше нормы де-
лителя. 
 Простейшим евклидовым кольцом является кольцо ℤ целых чисел (ев-
клидова норма совпадает с функцией "абсолютная величина": ݔ ↦  в ;(|ݔ|
этом кольце всего два обратимых элемента: ℤ∗ = {1, −1}.  
 К классу евклидовых относятся также все кольца ܲ[ݔ] многочленов, от 
одной переменной (univariate), с коэффициентами из любого поля ܲ; 
нормой служит функция "степень": ݂(ݔ) ⟼ deg (݂(ݔ)); обратимыми являют-
ся многочлены нулевой степени, т. е. ненулевые константы; мультиплика-
тивная группа кольца многочленов: (ܲ[ݔ])∗ = ܲ∗ = ܲ ∖ {0}.  
 Кольца многочленов от нескольких переменных (multivariate) уже 
не являются евклидовыми. Ниже будут приведены еще некоторые примеры 
евклидовых колец, возникающие в теории алгебраических чисел. 
 Над евклидовым кольцом ܭ любая (݉ × ݊)-матрица ܣ с помощью эле-
ментарных преобразований (трех типов, над строками и столбцами) может 
быть приведена к так называемой канонической форме Смита ܵ: по главной 
диагонали, в количестве, равном рангу ݎ = rank(ܣ) данной матрицы, распо-

лагаются ненулевые элементы ߤଵ, ,ଶߤ … , -причем каждый следу ,ܭ ௥ кольцаߤ
ющий элемент делится на предыдущий (ߤଵ| ߤଶ| …  ௥); остальные элементыߤ|
матрицы ܵ равны нулю. Форма Смита определена по матрице ܣ однозначно, 
с точностью до ассоциированности диагональных элементов (ассоциирован-
ными называются элементы, отличающиеся обратимым множителем). Эле-
менты ߤଵ, ,ଶߤ … ,   .ܣ ௥ называются инвариантными множителями дляߤ
 Форма Смита (для матриц над евклидовым кольцом) является обобще-
ние скелетного вида (для матриц над полем); в случае поля все инвариантные 
множители равны единице.  
 Матрицы ܣ и ܵ связаны соотношением ܵ = ݉) где ,ܸܣܷ × ݉)-матрица 
ܷ и (݊ × ݊)-матрица ܸ являются унимодулярными, т. е. их определители яв-
ляются обратимыми элементами (принадлежат ܭ∗); в матрице ܷ "накаплива-
ются" все элементарные преобразования над строками, а в матрице ܸ  - пре-
образования над столбцами. 
 Над кольцом ܭ = ℤ можно добиться полной однозначности в опреде-
лении формы Смита, если потребовать, чтобы инвариантные множители вы-
бирались положительными. Аналогично, для матриц над кольцом многочле-
нов ܭ = -форма Смита будет определена однозначно, если выбирать ин [ݔ]ܲ
вариантные множители (в данном случае, это - многочлены) нормализован-
ными, т. е. с единичным старшим коэффициентом (в "английском математи-
ческом" есть удобный короткий термин: 'monic polynomials'). 
  

Пример 1.3.1. Приведение целочисленной матрицы к канонической форме 
Смита.  
 

 Вычисления ведутся в (евклидовом) кольце целых чисел. 
 

sage: A=matrix(ZZ,[[0,-4,5,-3], 
                   [3,6,-12,6], 
                   [-9,-6,18,24]]);A 
 

[  0  -4   5  -3] 
[  3   6 -12   6] 
[ -9  -6  18  24]  
 

sage: S,U,V=A.smith_form(); S,U,V 
 

( 
                           [-4 -5 -6 16] 
[ 1  0  0  0]  [ 1  0  0]  [-4 -5 -5 13] 
[ 0  3  0  0]  [ 0  1  0]  [-3 -4 -4 11] 
[ 0  0 12  0], [-6 -1  1], [ 0  0  0  1] 
)  
 

 Форма Смита найдена; инвариантные множители матрицы 12 ,3 ,1 :ܣ. 
(Кстати, работа на эту тему опубликована английским математиком Генри 
Смитом в далеком 1861 г.; вот какие древние вещи мы изучаем.) 

 Далее следует проверка (напомним, что определители матриц перехода 
ܷ и ܸ должны принадлежать ℤ∗, т. е. равняться ±1). 
 

sage: U*A*V==S,U.det(),V.det() 
 

(True, 1, -1) 
 

Пример 1.3.2. Приведение полиномиальной матрицы к канонической форме 
Смита. 
 

 Вычисления ведутся в (евклидовом) кольце многочленов над полем ра-
циональных чисел. 
 

sage: x=var('x'); QQx.<x>=QQ[]; QQx 
 

Univariate Polynomial Ring in x over Rational Field 
 

A=matrix(QQx,[[x^2-2*x+1, -x^2+1, -2*x^2+2*x, 
 -x^3+x^2+x-1], [x^3-3*x^2+x+1, -x^4+x^3+2*x^2-x-1, 
 -3*x^4+4*x^3+7*x^2-6*x-2, x^2-1], [x^2-1, 
 x^4-x^3-4*x^2+x+3, 3*x^4-3*x^3-11*x^2+3*x+8, 
 -x^4+2*x^2-1]]) 
 

 Здесь и далее мы применяем искусственный прием "компактификации" 
визуального представления матриц, описанный в конце примера 1.2.2. 
 

m,n=A.nrows(),A.ncols(); m,n 
(3,4) 
 

[[A[i,j] for j in range(n)] for i in range(m)] 
 

[[x^2-2*x+1, -x^2+1, -2*x^2+2*x, -x^3+x^2+x-1], 
 [x^3-3*x^2+x+1, -x^4+x^3+2*x^2-x-1, 
  -3*x^4+4*x^3+7*x^2-6*x-2, x^2-1], 
 [x^2-1, x^4-x^3-4*x^2+x+3, 3*x^4-3*x^3-11*x^2+3*x+8, 
  -x^4+2*x^2-1]] 
 

S,U,V=A.smith_form() 
 

SS=[[S[i,j] for j in range(n)] for i in range(m)]; SS 
 

[[x-1, 0, 0, 0], [0, x^2-1, 0, 0], 
 [0, 0, -4*x^3+8*x^2+4*x-8, 0]] 
 

 Выделим из матрицы S список invar инвариантных множителей 
(многочленов), попутно разлагая их на неприводимые множители. 
 

r=S.rank() 
 

invar=[S[i,i].factor() for i in range(r)]; invar 
 

[x-1, (x-1)*(x+1), (-4)*(x-2)*(x-1)*(x+1)] 

 Наблюдаем, прежде всего, что требование делимости для инвариант-
ных множителей выполняется. В то же время замечаем, что третий из них не 
является нормализованным (старший коэффициент равен −4). По-видимому, 
алгоритм приведения к форме Смита над кольцом многочленов, встроенный 
в Sage, вообще не стремится к обеспечению нормализованности инвариант-
ных множителей, хотя ее легко достичь, например, с помощью домножения 
строк на (возможно дробные) константы. Теперь посмотрим на матрицы U 
и V и произведем проверку. 
 

UU=[[U[i,j] for j in range(m)] for i in range(m))]; UU 
 

[[1/2*x-1/2, -1/2, 0],  
 [1/2*x^3-x-1/2, -1/2*x^2+1/2, -1/2*x^2+1/2*x], 
 [x^4-2*x^3-2*x^2+5*x+2, -x^3+2*x^2+x-4, 
  -x^3+3*x^2-2*x-2]] 
 

VV=[[V[i,j] for j in range(n)] for i in range(n)]; VV 
 

[[1, -1/2*x^3+1/2*x^2+x, -x^3+x^2+2*x+1, x+3/2], 
 [0, 1, -1, -1/2], [0, 0, 1, 1/2], [0, 0, 0, 1]] 
 

S==U*A*V,U.det(),V.det() 
 

(True, 1, 1) 

2. Элементы общей (абстрактной) алгебры  
 

2.1. Поля, кольца, идеалы в кольцах, фактор-кольца 
 

 С важнейшими примерами числовых полей мы встреча-
лись в первой части пособия: QQ, RR, CC, QQbar, AA (в обозначениях 
Sage); с конечными полями познакомимся в п. 2.3. В первых двух пунктах 
второй части наши интересы будут сосредоточены на ассоциативных ком-
мутативных кольцах с единицей, а также на том, как посредством их факто-
ризации получаются поля.  
 Идеал ܬ в кольце ܭ определяется как подмножество в ܭ, являющееся 
подгруппой по сложению и устойчивое относительно умножения на элемен-
ты из ܭ (в коммутативном случае не важно, с какой стороны производится 
умножение и не различаются идеалы левые, правые и двусторонние). Кольцо 
-разбивается в объединение попарно не пересекающихся классов смежно ܭ
сти по идеалу ܬ, каждый из которых может быть представлен в виде ݔ +  ,ܬ
где ݔ - произвольный представитель класса. В реальных вычислениях, как 
правило, приходится фиксировать некоторый выделенный представитель 
в каждом из классов смежности. С помощью представителей корректным об-
разом определяются алгебраические действия над классами смежности: 
ݔ) + (ܬ + ݕ) + (ܬ = ݔ) + (ݕ + ݔ) ;ܬ + (ܬ ∙ ݕ) + (ܬ = ݔ) ∙ (ݕ + -тем самым мно ;ܬ
жество ܬ/ܭ всех классов смежности превращается в кольцо (называемое фак-
тор-кольцом (quotient ring) кольца ܭ по идеалу ܬ). 
 Особый и важнейший случай: если идеал ܬ является максимальным 
(т. е. не содержится ни в каком строго более широком идеале, отличном от 
всего кольца ܭ), то фактор-кольцо ܬ/ܭ оказывается полем. 
 

Пример 2.1.1. Кольцо целых чисел и кольца классов вычетов. 
 

 В кольце целых чисел ℤ  = ZZ всякий идеал является главным (может 
быть порожден одним элементом): ܬ = (݊) = ݊ℤ, где ݊ - неотрицательное це-
лое число; этот идеал является максимальным тогда и только тогда, когда 
число ݊ является простым натуральным; соответствующее фактор-кольцо 
обозначается ℤ௡ = ℤ/݊ℤ; в случае простого ݊ оно является полем. Sage про 
всё это знает; результат факторизации кодируется следующим образом: ℤ௡ = 
Intgers(n). 
 

sage: K=ZZ; K; J=K.ideal(8); J;  
 

Integer Ring 
Principal ideal (8) of Integer Ring 
 

 Определены кольцо ܭ и главный идеал ܬ, порожденный элементом 8. 
Ниже, с помощью метода .quotient_ring вычисляется фактор-кольцо 
ܮ =  тестируется ܮ и затем кольцо ܮ далее выводится список элементов ,ܬ/ܭ
на предмет того, является ли оно полем. 

sage: L=K.quotient_ring(J); L; L.list(); L.is_field() 
 

Ring of integers modulo 8 
[0, 1, 2, 3, 4, 5, 6, 7] 
False 
 Обратите внимание на то, что в качестве элементов фактор-кольца L 
фигурируют канонические представители классов смежности, а именно – 
наименьшие неотрицательные вычеты по модулю 8. 
 Доступны таблицы сложения и умножения для конечных колец. (Оп-
ция в скобках назначает в качестве имен элементов сами эти элементы; по 
умолчанию мы получили бы восемь первых букв английского алфавита.) 
 

sage: mt=L.multiplication_table(names='elements'); mt 
 

*  0 1 2 3 4 5 6 7 
 +---------------- 
0| 0 0 0 0 0 0 0 0 
1| 0 1 2 3 4 5 6 7 
2| 0 2 4 6 0 2 4 6 
3| 0 3 6 1 4 7 2 5 
4| 0 4 0 4 0 4 0 4 
5| 0 5 2 7 4 1 6 3 
6| 0 6 4 2 0 6 4 2 
7| 0 7 6 5 4 3 2 1 
 

 Из таблицы умножения усматриваем, что кольцо L не только не явля-
ется полем, но не является (в отличие от K) целостным кольцом (например, 
4 ∙ 6 = 0). Sage (реагируя на применение теста .is_integral_domain) 
подтверждает это наблюдение: 
 

sage: K.is_integral_domain(), L.is_integral_domain() 
 

(True, False) 
 Мы уже говорили о целостных кольцах в п. 1.3; упоминалось там и по-
нятие мультипликативной группы кольца, которая, по определению, состоит 
из обратимых элементов кольца, еще они называются единицами кольца. 
И здесь мы (используя тестирующий элементы на обратимость метод 
.is_unit) вычислим список элементов группы ℤ଼

∗  обратимых вычетов по 
модулю 8 
 

sage: U=[k for k in L if k.is_unit()]; U 
 

[1, 3, 5, 7] 
 

 Целостные кольца в некоторой степени "похожи" на кольцо целых чи-
сел (недоучившиеся алгебре пользователи могут даже путать эти термины). 
Однако более интересен более узкий класс колец – факториальные кольца.  

 Вот вам и новая возможность для путаницы! Факториальные коль-
ца - совсем не то же самое, что фактор-кольца.  
 Англоязычные компьютерщики не спутают: фактор-кольцо = quo-
tient ring, - это некоторое новое кольцо, которое строится по данному 
кольцу и идеалу в нем; факториальное кольцо = unique factorization 
domain (UFD), - это кольцо, в котором может быть развита содержательная 
теория делимости; дословный перевод: кольцо с однозначным разложением 
на множители. 
 Факториальные кольца "очень похожи" на кольцо целых чисел; в них, 
как и в ℤ, имеется "достаточно много" неразложимых (простых) элементов 
(каждый из которых не представим в виде произведения двух необратимых 
элементов), а всякий ненулевой и необратимый элемент - разлагается в про-
изведение неразложимых, причем такое разложение определено однозначно, 
с точностью до порядка сомножителей и их ассоциированности (напомним: 
ассоциированными называются элементы, отличающиеся обратимым мно-
жителем). 
 Класс факториальных колец является, однако, трудно обозримым; на 
практике используются некоторые достаточные условия факториальности. 
Например, доказано, что факториальными являются кольца главных идеалов 
(к.г.и. = PID, principal ideal domain), в которых (тоже, - как и в ℤ) 
всякий идеал является главным (см. определение в начале данного пункта). 
Sage - "в курсе", какие из знакомых ему колец являются к.г.и.; пользователи 
также должны понимать, что таковыми являются, в частности, все евклидовы 
кольца (е.к. = Euclidian domain; см. начало п. 1.3). 
 Следующий пример будет посвящен теме, к сожалению, не доведенной 
пока разработчиками Sage до "рабочего уровня", – вычислениям в кольцах 
алгебраических чисел.  
 

Пример 2.1.2. Алгебраические расширения поля рациональных чисел. 
 

Можно, присоединив мнимую единицу, расширить поле рациональных чисел 
ℚ  = QQ до так называемого поля рациональных гауссовых чисел (математи-
ческое обозначение: ℚ[݅]; на языке Sage следует указывать порождающий 
элемент).  
 

sage: GQ.<i>=QQ[I]; GQ; z=3/4+5/3*i; z; z in GQ; z^10 
 

Number Field in I with defining polynomial x^2+1 
5/3*I + 3/4 
True 
-316547951975/859963392*I + 11977223406401/61917364224  
 

 Кольцо ℤ[݅] целых гауссовых чисел может быть получено как кольцо 
целых элементов поля ℚ[݅].  
 

sage: GZ=GQ.ring_of_integers(); GZ; GZ.gens(); z in GZ 

Maximal Order in Number Field in I with defining  
 polynomial x^2+1 
[1, I] 
False 
 

 Над целыми и рациональными гауссовыми числами можно выполнять 
простейшие алгебраические действия, однако разлагать гауссовы целые чис-
ла на простые (неразложимые) множители Sage пока не умеет.  
 Есть над чем работать! Можно надеяться, что энергичным неофитам 
окажется по силам довести технику Sage в этом частном вопросе хотя бы до 
уровня Maple. Почитайте предварительно теорию: в учебниках по алгебре 
можно найти доказательство евклидовости (и, следовательно, факториаль-
ности) кольца ℤ[݅] роль евклидовой нормы играет квадрат модуля комплекс-
ного числа; обратимых элементов в этом кольце четыре: ±1 и  ±݅; неразло-
жимые элементы распознаются так: это, во-первых, четыре числа с нормой 2, 
а именно: ±1 ± ݅, во-вторых, все числа, норма которых равна нечетному про-
стому числу, и, в-третьих, числа, норма которых равна квадрату простого 
числа, но не любого, а сравнимого с тройкой по модулю 4. 
 Интересно, что Sage все же способен решать задачу разложения на 
множители, но не "на уровне элементов", а на уровне так называемых дроб-
ных идеалов. К сожалению, уровень ожидаемой математической подготовки 
читателей данного пособия не дает нам возможности углубляться в такие 
"дебри". 
 Покажем еще поле рациональных и кольцо целых чисел Эйзенштейна. 
Поле ℚ[݅√3], получается присоединением к полю ℚ элемента √−3 = ݅√3 
(являющегося корнем многочлена ݔଶ + 3; по умолчанию Sage использует 
для добавляемого элемента обозначение  a; поле расширения состоит из все-
возможных линейных комбинаций вида x*1+y*a с рациональными коэффи-
циентами x,y).  
 

sage: EQ=QQ[sqrt(-3)]; EQ 
 

Number Field in a with defining polynomial x^2+3 
 

 Однако, в какой-то степени неожиданно, оказывается, что подкольцо 
целых элементов поля ℚ[݅√3] состоит из линейных комбинаций такого же 
вида, x*1+y*a, однако (вопреки ожиданиям) не обязательно - с целыми 
x,y; эти коэффициенты могут быть либо оба целыми, либо оба полуцелыми. 
Sage знает об этом. 
 

sage: EZ=EQ.ring_of_integers(); EZ; w,v=EZ.gens(); w,v 
 

Maximal Order in Number Field in a with defining  
 polynomial x^2+3 
(1/2*a+1/2, a) 

 Среди порождающих элементов (генераторов) кольца оказалось число 
଺ߞ = ଵ

ଶ
(1 + ܽ) = ଵ

ଶ
(1 + ݅√3), которое есть не что иное, как первообразный 

комплексный корень шестой степени из единицы. Почему же такое число, с 
нецелыми коэффициентами, заслуживает название целого? А по определе-
нию! Вспомним, что алгебраическое число – это корень многочлена с целыми 
коэффициентами, а целое алгебраическое число – корень нормализованного 
многочлена с целыми коэффициентами. Ниже мы тестируем число ݓ =  ଺ наߞ
принадлежность полю алгебраических чисел, а потом находим для этого чис-
ла минимальный аннулирующий многочлен (т. е. многочлен наименьшей 
степени с целыми или рациональными коэффициентами, корнем которого 
является ݓ).  
 

sage: w=(1+I*sqrt(3))/2; w; w in QQbar; w.minpoly() 
 

1/2*I*sqrt(3)+1/2 
True 
x^2–x+1 
 

 Минимальный многочлен имеет целые коэффициенты и является нор-
мализованным. Значит, w является целым алгебраическим числом. Повторим 
испытание для другого числа. 
 

sage: w1=(2+I*sqrt(3))/2; w1; w1 in QQbar; w1.minpoly() 
 

1/2*I*sqrt(3)+1 
True 
x^2-2*x+7/4 
 

 Выдан нормализованный минимальный многочлен, но – с рациональ-
ными коэффициентами. Коэффициенты можно сделать целыми, если домно-
жить многочлен на 4, но тогда он перестанет быть нормализованным. Число 
w1 является алгебраическим, но не целым алгебраическим. 
 Нетрудно заметить, что всякое целое число Эйзенштейна можно пред-
ставить также целочисленной линейной комбинацией несколько иного вида: 
ݑ ∙ 1 + ݒ ∙  ଷ, в которой участвует первообразный корень третьей степени изߞ
единицы ߞଷ = ଵ

ଶ
(−1 + ݅√3). Поэтому математически корректным обозначе-

нием кольца целых чисел Эйзенштейна может служить ℤ[ߞଷ]. Кольцо целых 
чисел Эйзенштейна также является е.к., норма опять же определяется как 
квадрат модуля; обратимые элементы образуют циклическую группу шесто-
го порядка, порожденную ߞ଺; описание неразложимых элементов не будем 
здесь приводить, укажем только, что, вместо сравнений по модулю 4, прихо-
дится использовать сравнения по модулю 3). 

2.2. Кольца многочленов и фактор-кольца для них  
 

 С тем, как определяются в системе Sage кольца от одной и 
нескольких переменных, мы знакомились с первых же страниц 
пособия (см. пример 1.1.1), причем рассматривались кольца 
многочленов как над полям, так и над другими кольцами. Кольцо ܭ =  [ݔ]ܲ
многочленов от одной переменной над произвольным полем ܲ является важ-
нейшим примером евклидова кольца; в частности, в нем все идеалы являются 
главными (principal) и справедлива теорема об однозначном (с точностью 
до ассоциированности) разложении на множители. Напомним, что группа 
обратимых многочленов состоит из ненулевых констант, ассоциирован-
ность, в данном случае, - это пропорциональность; для любых двух много-
членов определены (причем, однозначно, если их выбирать нормализован-
ными) наибольший общий делитель (НОД = gcd, greatest common 
divisor) и наименьшее общее кратное (НОК = lcm, least common 
multiple); неразложимые элементы в кольце многочленов принято назы-
вать неприводимыми (irreducible) многочленами. А теперь мы все это 
покажем средствами Sage. 
 

Пример 2.2.1. Теория делимости в кольце многочленов над полем. 
 

sage: QQx=PolynomialRing(QQ,n,'x'); QQx 
 

Univariate Polynomial Ring in x over Rational Field 
 

 Введем два многочлена с целыми коэффициентами (но рассматривают-
ся они над QQ); поделим с остатком второй многочлен на первый (quo = 
quotient, неполное частное, rem = remainder, остаток); выполним 
проверку; затем вычислим НОД и НОК, после чего реализуем расширенный 
алгоритм Евклида: вычисляется не только НОД, но и линейное представле-
ние для него (в обозначении метода .xgcd префикс x отсылает к термину 
extended). 
 

sage: f= 2*x^5+5*x^3+x^2+2*x+2; 
      g=3*x^7+6*x^5-x^3+x^2-2*x+2; f; g 
 

2*x^5+5*x^3+x^2+2*x+2 
3*x^7+6*x^5-x^3+x^2-2*x+2 
 

sage: q,r=g.quo_rem(f); q; r; q*f+r==g 
 

3/2*x^2-3/4 
-3/2*x^4-1/4*x^3-5/4*x^2-1/2*x+7/2 
True 

sage: f.gcd(g); f.lcm(g) 
 

x^2+2 
x^10+5/2*x^8+1/2*x^7+2/3*x^6+4/3*x^5-5/6*x^4+2/3*x^3 
 -1/6*x^2+1/3 
 

sage: d,u,v=f.xgcd(g); d; u; v; f*u+g*v==d 
 

x^2+2 
-183/361*x^4+33/361*x^3-219/361*x^2+75/361*x+154/361 
122/361*x^2-22/361*x+207/361 
True 
 

 Далее введем еще один многочлен (над полем QQ) и произведем фак-
торизацию, т. е. разложим этот многочлен на неприводимые множители; по-
кажем, как извлекаются из объекта факторизации отдельные компоненты: 
пары вида  
(неприводимый множитель, его кратность в разложении), 

а также сами неприводимые множители. 
 

sage:  h=2*x^5+5/2*x^4+3/4*x^3-25/24*x^2-x-1/2; h 
 

2*x^5+5/2*x^4+3/4*x^3-25/24*x^2-x-1/2 
 

sage:  h.is_irreducible() 
 

False 
 

sage: hf=h.factor(); hf 
 

(2)*(x^2+3/2*x+3/4)*(x^3-1/4*x^2-1/3) 
 

sage: hf[0]; hf[1] 
 

(x^2+3/2*x+3/4, 1) 
(x^3-1/4*x^2-1/3, 1) 
 

sage: p=hf[1][0]; p; p.is_irreducible() 
 

x^3-1/4*x^2-1/3 
True 
 

 Напомним, что над алгебраически замкнутым полем неприводимыми 
многочленами являются многочлены первой степени и только они. 
 



27 
 

sage: p.change_ring(QQbar).factor() 
 

(x-0.7874995561234334?)* 
(x+0.2687497780617167?-0.5924982857827728?*I)* 
(x+0.2687497780617167?+0.5924982857827728?*I)  
 

Пример 2.2.2. Фактор-кольца кольца многочленов. 
 

 А теперь займемся факторизацией "в другом смысле". Первый смысл: 
факторизация – это разложение чего-либо на множители (факторы); второй 
смысл: факторизация – это построение по некоторому алгебраическому объ-
екту и его подобъекту нового объекта, называемого фактор-объектом. Кон-
кретно, сейчас речь пойдет о вычислении фактор-кольца для кольца много-
членов по некоторому идеалу в этом кольце (аналогичной задачей мы уже за-
нимались в п. 2.1, но – в более простом кольце ZZ). 
 Кольцо многочленов будет рассматриваться то же самое: QQx; сначала 
мы определим в нем три идеала; первые два будут главными по построению, 
третий, по построению, будет порождаться двумя многочленами, но и он 
окажется главным: порождающим для него будет НОД двух исходных мно-
гочленов; все три идеала будут протестированы на максимальность. Затем 
мы вычислим три фактор-кольца, и Sage подтвердит наши познания: фак-
тор-кольцо по некоторому идеалу является полем тогда и только тогда, когда 
этот идеал является максимальным, что, для главных идеалов, равносильно 
неприводимости порождающего идеал многочлена. 
 

sage: f=x^2+1; J1=QQx.ideal(f); J1; J1.is_maximal() 
 

Principal ideal (x^2+1) of Univariate Polynomial Ring 
 in x over Rational Field 
True 
 

sage: g=x^3+1; J2=QQx.ideal(g); J2; J2.is_maximal() 
 

Principal ideal (x^3+1) of Univariate Polynomial Ring 
 in x over Rational Field 
False 
 

sage: J3=QQx.ideal(x^4-1,x^3-x); J3; J3.is_maximal() 
 

Principal ideal (x^2-1) of Univariate Polynomial Ring 
 in x over Rational Field 
False 

 Далее мы вычисли три фактор-кольца. В первом из них вы обязаны 
"опознать" ранее уже встречавшееся нам (см. пример 2.1.2) поле гауссовых 
чисел QQ[I]. Обратите также внимание на метод выявления порождающего 
элемента (генератора) для фактор-кольца; по умолчанию он получает имя 
xbar (но может быть переименован, мы показываем как); через него выра-
жаются все элементы фактор-кольца и, в частности, генерируемые далее слу-
чайные элементы). 
 

sage: K1=QQx.quotient_ring(J1); K1; K1.is_field(); 
      K1.0; K1.random_element() 
 

Univariate Quotient Polynomial Ring in xbar over 
 Rational Field with modulus x^2+1 
True 
xbar 
-16*xbar-1/5 
 

sage: K2.<y>=QQx.quotient_ring(J2); K2; K2.is_field(); 
      K2.0; K2.random_element() 
 

Univariate Quotient Polynomial Ring in y over Rational 
 Field with modulus x^3+1 
False 
y 
-1/2*y^2-1/6*y+3 
 

sage: K3.<z>=QQx.quotient_ring(J3); K3; K3.is_field(); 
      K3.0; K3.random_element() 
 

Univariate Quotient Polynomial Ring in z over Rational 
 Field with modulus x^2-1 
False 
z 
3/5*z+1 
 

 Поясним теперь термин modulus, используемый Sage при описании 
фактор-колец. Надо полагать, что к "модулярным вычислениям" в кольцах 
вычетов читатели уже приучены. Например, в фактор-кольце ℤଵଶ = ℤ ∕ (12) 
все вычисления ведутся "по модулю 12", т. е. сложение и умножение вычетов 
производятся как над обычными целыми числами, с последующей заменой 
результата на остаток от его деления на модуль. Аналогично обстоит дело и 

в фактор-кольцах кольца многочленов, но теперь modulus - это многочлен, 
и именно на него надо делить с остатком. Sage с этим справляется. Пример 
приведем в кольце K2. 
 

sage: a=3/11*y^2-5*y+4/3; a; a in K2;  
      b=-2*y^2-1/7; b; b in K2; 
      a*b 
 

3/11*y^2-5*y+4/3 
True 
-2*y^2-1/7 
True 
-625/231*y^2+97/77*y-214/21 
 

 В заключение пункта остается посетовать на известную (и, по-видимо-
му, неустранимую) особенность всевозможных учебных пособий: тему при-
ходится "закруглять" именно тогда, когда мы приближаемся к самому инте-
ресному и интригующему. Самым интригующим в теме "Многочлены" явля-
ется теория идеалов в кольцах многочленов от нескольких переменных, кото-
рая, в свою очередь, служит основой для построения алгоритмов решения не-
линейных систем уравнений. Автор надеется, что читатели данного пособия 
(на старших курсах, или даже в послеуниверситетской математической жиз-
ни) непременно встретятся с понятием базис Грёбнера (для полиномиального 
идеала), научатся такие базисы строить и применять к реальным (и пото-
му - сложным) задачам. С основами теории базисов Грёбнера и с соответ-
ствующими возможностями Sage можно познакомиться, обратившись к ре-
комендованной литературе (плюс сетевой поиск). 

2.3. Конечные поля  
 

 Поле называется простым, если оно не содержит соб-
ственных подполей. Совсем просто доказывается, что един-
ственными (с точностью до изоморфизма) простыми полями яв-
ляются: (1) поле рациональных чисел ℚ; (2) поля ॲ௣ = ℤ௣ классов вычетов по 
произвольному простому модулю ݌. 
 Произвольное поле ܨ обязано содержать минимальное (простое) под-
поле. Если для любого натурального ݊ сумма ݊ штук полевых единиц (обо-
значаемая ݊1) отлична от нуля, то минимальное подполе в поле ܨ изоморф-
но ℚ; в данном случае принято говорить, что поле ܨ имеет нулевую характе-
ристику (char(ܨ) = 0), при этом оно обязательно бесконечно.  
 Если среди элементов ݊1 встречаются нулевые, то наименьшее из со-
ответствующих ݊ непременно оказывается простым (переобозначим: ݊ =  ,(݌
тогда минимальное подполе в ܨ будет изоморфно ॲ௣ и, по определению, счи-
тается: char(ܨ) = -в данной ситуации может быть как конеч ܨ само поле ;݌
ным, так и бесконечным. 
 Всякое конечное поле ܨ имеет ненулевую характеристику (char(ܨ) =  ݌
для некоторого простого ݌) и оказывается линейным пространством (неко-
торой конечной размерности ݏ) над своим простым подполем. Отсюда немед-
ленно следует, что количество элементов (порядок) ݍ конечного поля обяза-
тельно является примарным числом (т. е. степенью простого числа): ݍ =  .௦݌
Доказывается, что для любого примарного числа ݍ = -௦ существует един݌
ственное (с точностью до изоморфизма) поле порядка  ݍ; оно называется по-
лем Галуа (Galois Field); наряду с обозначением ॲ௤, используются вари-
анты: (ݍ)ܨܩ или ݌)ܨܩ,  .(ݏ
 Практическая конструкция поля Галуа ॲ௤ (ݍ = -௦) основана на отыска݌
нии некоторого неприводимого многочлена ݂(ݔ) степени ݏ над простым по-
лем ॲ௣, с последующей факторизацией (см. пример 2.2.2) кольца многочле-
нов ܭ = ॲ௣[ݔ] по главному идеалу ܬ =   .((ݔ)݂)
 Фактор-кольцо ܬ/ܭ является, благодаря неприводимости ݂(ݔ), полем; 
оно одержит ровно ݍ элементов (именно столько многочленов степени, не 
превышающей ݏ − 1, существует над полем ॲ௣; эти многочлены принимают-
ся в качестве канонических представителей классов смежности по идеалу ܬ).  
 Для обеспечения результативности и корректности описанной кон-
струкции требуется предварительно убедиться в том, что 
 (1) над полем ॲ௣ существуют неприводимые многочлены произвольной 
степени ݏ; 
 (2) при различных выборах неприводимого многочлена одной и той же 
степени ݏ при факторизации получаются изоморфные поля.  
 Эти и все другие подробности теории читателям следует изучить по ре-
комендуемой учебной литературе. Здесь же мы можем подвести предвари-
тельный итог.  

 Конструкция  
ࡲ = ॲ((࢞)ࢌ)/[࢞]࢖    (*)  

при произвольном выборе неприводимого многочлена ࢌ(࢞) степени ࢙ при-
водит к полю ࡲ, изоморфному ॲ࢙࢖ . 
 
 

 Однако это еще не все трудности, которые необходимо преодолеть. 
Есть, по крайней мере, еще две серьезные проблемы.  
 Первая такова: откуда брать неприводимые многочлены? Не отвлекаясь 
на (вообще говоря, важные) детали сформулируем критерий.  
 
 

 Нормализованный многочлен ࢌ(࢞) степени ࢙ над полем ॲ࢖ неприво-
дим тогда и только тогда, когда выполнено "условие делимости": 

૚ି࢙࢖࢞ | (࢞)ࢌ − ૚.    (**)  
 
 

 Вторая проблема: как представлять элементы построенного поля? Дело 
в том, что данное выше их описание с помощью многочленов степени не 
выше ݏ − 1 не всегда оказывается "удачным". При хорошем описании все 
ненулевые элементы поля должны представляться в виде степеней одного из 
них (генератора). В данном случае хотелось бы, чтобы генератором послу-
жил многочлен первой степени ݔ. Но это не всегда получается. (Какой-либо 
генератор существует всегда, но другие генераторы менее удобны.)  
 К счастью, имеется выход. Назовем неприводимый многочлен ݂(ݔ) 
примитивным, если при факторизации (*) одночлен ݔ является генератором 
поля. Доказывается, что для любого ݏ существуют примитивные многочлены 
степени ݏ и формулируется следующий критерий примитивности. 
 
 

 Нормализованный многочлен ࢌ(࢞) степени ࢙ над полем ॲ࢖ является 
примитивным тогда и только тогда, когда, помимо условия делимости 
(**), выполнено следующее "условие неделимости": для любого нату-
рального показателя ࢑ < ࢙࢖ − ૚ справедливо: 

(࢞)ࢌ  ∤  ࢞࢑ − ૚.     (***)  
 
 

 Известно, что (***) достаточно проверять для ݇ | ݌௦ − 1. Это известно 
не только математикам, но и их изобретению – системе Sage. В частности, в 
Sage реализован алгоритм вычисления так называемых примитивных поли-
номов Конвея, которые используются при "стандартном" задании системой 
полей Галуа.) 
 И, наконец, еще один ключевой момент в теории конечных полей. 

 Мультипликативная группа ॲࢗ
∗ = ॲࢗ ∖ {૙} поля Галуа является цик-

лической группой порядка ࢗ − ૚; в случае примитивности многочлена 
 в качестве порождающего элемента этой группы можно принять (࢞)ࢌ
класс смежности одночлена ࢞, т. е. всякий элемент ࢟ ∈ ॲࢗ

∗  представля-
ется в виде 

࢟ = ࢞࢑,    (****)  
где ࢑ - целое число (оно определено однозначно, если предполагать, что 
૙ ≤ ࢑ ≤ ࢗ − ૛). 
 
 

 Формула (****) называется степенным представлением для (ненуле-
вых) элементов конечного поля. 
 

 Теперь пора переходить к примерам. Начнем не с полей, а с многочле-
нов, которые при построении полей используются. 
 

Пример 2.3.1. Полиномы Конвея. 
 

 Строгое определение полиномов Конвея (Conway) вы можете найти, 
например, в Википедии. Здесь мы покажем только способ извлечения нужно-
го многочлена из "хранилища" Sage, выявим для него "родительское коль-
цо" (метод .parent), после чего применим методы тестирования на непри-
водимость и примитивность. 
 

sage: f=conway_polynomial(2,4); f; f.parent(); 
      f.is_irreducible(),f.is_primitive() 
 

x^4+x+1 
Univariate Polynomial Ring in x over Finite Field of 
 size 2 (using NTL) 
(True, True) 
 

sage: f=conway_polynomial(5,3); f; f.parent(); 
  f.is_irreducible(),f.is_primitive() 
 

x^3 + 3*x + 3 
Univariate Polynomial Ring in x over Finite Field of 
 size 5 
(True, True) 
 

Пример 2.3.2. Поле Галуа порядка 8. 
 

 Зададим поле из восьми элементов F8 = ॲ଼ = ॲଶ[ݔ]/(݂(ݔ)) [см. схе-
му (*)] и выведем список его элементов; все они выражаются через генера-
тор, который в строго математическом смысле является классом смежности 
ݔ + -однако, по нашему выбору (в угловых скобках), обозначатся про ,((ݔ)݂)

сто x. Можно тут же запросить используемый системой многочлен ݂(ݔ) и 
убедиться в том, что он совпадает с соответствующим полиномом Конвея. 
 

sage: F8.<x>=GF(2^3); L=F8.list(); L;  
      f=F8.polynomial(); f;  
      f==conway_polynomial(2,3) 
 

[0, x, x^2, x+1, x^2+x, x^2+x+1, x^2+1, 1] 
x^3+x+1 
True 
 

 Обратите внимание на порядок элементов в списке. 
 Естественным можно было бы считать такой порядок в множестве мно-
гочленов, при котором они группируются по возрастанию степени, а внутри 
групп – упорядочиваются лексикографически, с привлечением обычного по-
рядка [0,1,2,…,p-1] в поле ॲ௣. По такому принципу в нашем примере 
получилось бы: [0, 1, x, x+1, x^2, x^2+1, x^2+x, x^2+x+1].  
 Чем мотивирован порядок, принятый Sage, будет объяснено далее. 
 А сейчас мы (в продолжение замечания о "модулярных вычислениях" в 
конце примера 2.2.2) подчеркнем следующее обстоятельство: вычисления в 
конечных полях также являются модулярными, роль modulus играет тот 
самый многочлен ݂(ݔ), с помощью которого определялось поле.  
 Но в данном случае ситуация еще интереснее, поскольку мы фактиче-
ски работаем с двумя модулями: во-первых, коэффициенты всех задейство-
ванных многочленов рассматриваются по модулю простого числа ݌, а, во-
вторых, сами многочлены приводятся по модулю (неприводимого и прими-
тивного) многочлена ݂(ݔ).  
 Доступны полевые таблицы сложения и умножения. По умолчанию 
элементам поля присвоены буквенные обозначения: восемь первых букв ла-
тинского алфавита кодируют восемь элементов поля, причем именно в том 
порядке, в котором они занесены выше в список L.  
 Представим здесь таблицу умножения для ॲ଼ и сравним ее с получен-
ной ранее, в примере 2.1.1, таблицей умножения для ℤ଼ (в которой, для еди-
нообразия, сменим кодировку элементов на буквенную).  
 Попробуйте понять, в чем принципиальное различие между этими таб-
лицами? 
 

sage: F8.multiplication_table(), 
      Z8.multiplication_table() 

( 
*  a b c d e f g h    *  a b c d e f g h 
 +----------------     +---------------- 
a| a a a a a a a a    a| a a a a a a a a 
b| a c d e f g h b    b| a b c d e f g h 
c| a d e f g h b c    c| a c e g a c e g 
d| a e f g h b c d    d| a d g b e h c f 
e| a f g h b c d e    e| a e a e a e a e 
f| a g h b c d e f    f| a f c h e b g d 
g| a h b c d e f g    g| a g e c a g e c 
h| a b c d e f g h ,  h| a h g f e d c b 
) 
 

 Читателям следует четко осознать, что ℤ௤ (кольцо), при составных ݍ, 
отнюдь не идентично с ॲ௤ (полем); тогда как для простых значений порядка 
ݍ) = можно считать, что ॲ௣ (݌ = ℤ௣. 
 Далее поупражняемся в арифметических действиях над элементами 
поля. 
 

sage: x^3, x^1000, 1/x, (1+x^3+x^94)/(1+x^4+x^80) 
 

(x+1, x^2+1, x^2+1, x^2+x+1) 
 

 И, наконец, вернемся к якобы "неестественной" нумерации элементов 
поля ॲ௤. На самом деле она достаточно естественна: в качестве номера эле-
мента берется логарифм этого это элемента по основанию, равному генера-
тору поля. Точнее, благодаря степенному представлению (****), всякий эле-
мент ݕ ∈ ॲ௤

∗  представляется в виде ݕ = ௞ݔ , где ݔ - генератор. Число ݇ есте-
ственно назвать логарифмом элемента ݕ по основанию ݔ; используется 
обычное обозначение: ݇ = log௫ -Логарифм определен однозначно по моду .ݕ
лю ݍ − 1, и – совсем однозначно, если ограничиться значениями от 0 до 
ݍ − 2. Чтобы не исключать из нумерации нулевой элемент, производится не-
которая модификация: номер единичного элемента увеличивается на период, 
т. е. становится равным ݍ − 1, а номер 0 закрепляется за нулевым элементом. 
 

 Приведем, без подробных комментариев, следующий 
 

Пример 2.3.3. Поле Галуа порядка 81. 
 

sage: F81.<x>=GF(3^4); L=F81.list(); L;  
      f=F81.polynomial(); f;  

[0, x, x^2, x^3, x^3+1, x^3+x+1, x^3+x^2+x+1, 
 2*x^3+x^2+x+1, x^2+x+2, x^3+x^2+2*x, 2*x^3+2*x^2+1, 
 x^3+x+2, x^3+x^2+2*x+1, 2*x^3+2*x^2+x+1, x^3+x^2+x+2, 
 2*x^3+x^2+2*x+1, 2*x^2+x+2, 2*x^3+x^2+2*x, 2*x^2+2, 
 2*x^3+2*x, 2*x^3+2*x^2+2, x^3+2*x+2, x^3+2*x^2+2*x+1, 
 2*x^2+x+1, 2*x^3+x^2+x, x^2+2, x^3+2*x, x^3+2*x^2+1, 
 x+1, x^2+x, x^3+x^2, 2*x^3+1, 2*x^3+x+2, 
 2*x^3+x^2+2*x+2, 2*x^2+2*x+2, 2*x^3+2*x^2+2*x, 
 x^3+2*x^2+2, 2*x+1, 2*x^2+x, 2*x^3+x^2, 2, 2*x, 2*x^2, 
 2*x^3, 2*x^3+2, 2*x^3+2*x+2, 2*x^3+2*x^2+2*x+2, 
 x^3+2*x^2+2*x+2, 2*x^2+2*x+1, 2*x^3+2*x^2+x, 
 x^3+x^2+2, 2*x^3+2*x+1, 2*x^3+2*x^2+x+2, 
 x^3+x^2+2*x+2, 2*x^3+2*x^2+2*x+1, x^3+2*x^2+x+2, 
 x^2+2*x+1, x^3+2*x^2+x, x^2+1, x^3+x, x^3+x^2+1, 
 2*x^3+x+1, 2*x^3+x^2+x+2, x^2+2*x+2, x^3+2*x^2+2*x, 
 2*x^2+1, 2*x^3+x, 2*x^3+x^2+2, 2*x+2, 2*x^2+2*x, 
 2*x^3+2*x^2, x^3+2, x^3+2*x+1, x^3+2*x^2+x+1, x^2+x+1, 
 x^3+x^2+x, 2*x^3+x^2+1, x+2, x^2+2*x, x^3+2*x^2, 1] 
 

x^4 + 2*x^3 + 2 
 

sage: y=2*x^3+x^2+2; y; k=y.log(x); k; y==L[k] 
2*x^3 + x^2 + 2 
67 
True 
 

sage: (L[11]*L[68]).log(x) 
79 
 Надо полагать, что читатели обратили внимание, на, может быть, не 
совсем привычный синтаксис метода .log: в скобках, после знака логариф-
ма, указывается основание логарифмов, аргумент ставится перед точкой и 
знаком функции. Важно также, что степенное представление делает легко 
осуществимым умножение элементов: при умножении показатели склады-
ваются по модулю ݌. Разумеется, за это приходится платить усложнением ал-
горитма сложения. Что касается операции обращения элементов, то, в силу 
периодичности значений степеней, ее можно заменить вычислением подхо-
дящей степени: x^(q-1)=1 влечет x^(-1)=x^(q-2), и, более общим об-
разом, L[k]^(-1)=L[q-k-1]. 

2.4. Группы  
 

 До сих пор почти все рассматривавшиеся нами Sage-
объекты и операции относились к коммутативной алгебре. 
(Единственным исключением можно считать алгебру квадрат-
ных матриц заданного размера.) Теперь же мы обращаемся к изучению групп, 
которые могут быть как коммутативными, так и некоммутативными. Группа 
является, в принципе, более простым и фундаментальным объектом, нежели 
кольцо или поле, поскольку на ней задана всего лишь одна основная алгебра-
ическая операция (сложение или умножение), а не две взаимодействующих 
операции (как в случае колец). С другой стороны, теория групп значительно 
богаче важными примерами и значительно глубже развита. В связи с этим 
"групповая" техника в Sage также является более сложной, продвинутой. 
Мы сможем здесь остановиться лишь на простейших типах и примерах (в ос-
новном, конечных) групп. 
 

Пример 2.4.1. Группы перестановок. 
 

 Перестановки мы понимаем как биекции конечного множества. В учеб-
никах принята двустрочная запись перестановок, явно указывающая, какой 
элемент куда переходит.  
 Перестановки, будучи отображениями и умножаются как отображе-
ния, т. е. произведение перестановок – это композиция. В большинстве со-
временных учебников по алгебре принято выполнять композицию справа 
налево, и это правило распространяется на перестановки. Однако сохраняет-
ся и противоположная традиция. Во многих классических руководствах учат 
перемножать перестановки слева направо (хотя в соседних параграфах дру-
гие отображения перемножаются в противоположном порядке). Этот "разно-
бой" был привнесен и в Sage, о чем сожалеет, в частности, один из активных 
пропагандистов системы R. Beezer (см. [4, 5]). Пока с этим ничего поделать 
нельзя, просто читатели должны помнить, что в данном пособии произведе-
ние перестановок выполняется не так, как, скажем, в [13] или [20], а – в про-
тивоположном порядке.  
 Для кодирования перестановок в системах компьютерной алгебры дву-
строчная запись, как правило, не используется, ввиду ее избыточности. Го-
раздо более экономной оказывается задание перестановки ее разложением на 
независимые циклы. Именно его применяет Sage, причем в двух версиях: 
с привлечением строчного типа (string) и в виде списка кортежей 
(tuples). Мы пойдем по второму пути; подскажем только тем пользовате-
лям, который захотят использовать циклы длины единица (делать это без 

особой нужды, разумеется, не стоит), что в круглых скобках, после един-
ственного элемента кортежа единичной длины, полагается запятая.  
 По записи перестановки с помощью разложения на циклы степень пе-
рестановки однозначно не определяется, поэтому ввод перестановок обычно 
предваряется заданием объемлющей группы, которая имеет стандартное имя: 
симметрическая. 
 

sage: S6=SymmetricGroup(6); S6 
 

Symmetric group of order 6! as a permutation group 
 

sage: epsilon=S6([()]); epsilon; epsilon in S6; 
      phi=S6([(1,),(4,2,6),(5,3)]); phi; phi in S6; 
      psi=S6([(6,2,3),(5,4)]); psi; psi in S6 
 

() 
True 
(2,6,4)(3,5) 
True 
(2,6)(4,5) 
True 
 

 Sage аккуратен: он опознал, тождественную перестановку, убрал не-
нужный тривиальный цикл, во всех нетривиальных циклах переделал запись 
так, чтобы цикл начинался с минимального по номеру элемента, и (внима-
ние!) "ликвидировал" характерные для списков квадратные скобки и запятые 
и тем самым фактически перешел от списочной к строчной записи переста-
новок. Далее мы производим перемножение элементов и наблюдаем неком-
мутативность. 
 

sage: alpha=phi*psi; alpha; beta=psi*phi; beta; 
      alpha==beta 
 

(3,4,6,5) 
(2,4,3,5) 
False 
 А теперь мы поработаем с группой ܵସ перестановок степени 4: выведем 
список всех ее элементов, определим порядок (мощность) группы, протести-
руем группу на цикличность и коммутативность, вычислим порядки всех ее 
элементов, найдем порождающие элементы (generators). [Всякий эле-
мент группы может быть представлен в виде произведения генераторов и 
элементов, обратных к генераторам.] 

sage: S4=SymmetricGroup(4); S4; S4.order(); 
      LS4=S4.list(); LS4 
 

Symmetric group of order 4! as a permutation group 
24 
[(),(3,4),(2,3),(2,3,4),(2,4,3),(2,4),(1,2),(1,2)(3,4), 
 (1,2,3),(1,2,3,4),(1,2,4,3),(1,2,4),(1,3,2),(1,3,4,2), 
 (1,3),(1,3,4),(1,3)(2,4),(1,3,2,4),(1,4,3,2),(1,4,2), 
 (1,4,3),(1,4),(1,4,2,3),(1,4)(2,3)] 
 

sage: S4.is_cyclic(); S4.is_abelian();  
      [k.order() for k in S4] 
 

False 
False 
[1,2,2,3,3,2,2,2,3,4,4,3,3,4,2,3,2,4,4,3,3,2,4,2] 
 

sage: a,b=S4.gens(); a,b 
 

((1,2,3,4),(1,2)) 
 

 Перейдем к исследованию некоторых известных подгрупп в группе ܵସ 
и начнем с диэдральной группы ܦସ, являющейся группой симметрий квадра-
та. Убедимся, что она действительно является подгруппой в группе ܵସ (не-
циклической, некоммутативной и ненормальной), определим генераторы.  
 

sage: D4=DihedralGroup(4); D4; D4.order(); 
      LD4=D4.list(); LD4 
 

Dihedral group of order 8 as a permutation group 
8 
[(),(2,4),(1,2)(3,4),(1,2,3,4),(1,3),(1,3)(2,4), 
 (1,4,3,2),(1,4)(2,3)] 
 

sage: D4.is_subgroup(S4), D4.is_cyclic(), 
      D4.is_abelian(), D4.is_normal(S4); D4.gens() 
 

(True,False,False,False) 
[(1,2,3,4),(1,4)(2,3)] 
 

 Итак, диэдральная группа ܦସ является подгруппой, хотя и не нормаль-
ной, индекса 24/8 = 3 в симметрической группе ܵସ. Следовательно, мы имеем 
два (равномощных, но различных) фактор-множества: ܵସ ∕ ସܦ ସ иܦ ∖ ܵସ; пер-



39 
 

вое из них состоит из классов смежности (= cosets) вида ܦݔସ, в количе-
стве, равном индексу, а второе – из классов смежности вида ܦସݔ, в таком же 
количестве. Здесь снова "в товарищах согласья нет": которые из этих классов 
левые, а которые – правые? Обычно левыми называются классы первого типа 
(в них представитель ݔ располагается слева от подгруппы, в правых клас-
сах - наоборот). Sage снова "путает" ориентацию (меняет на противополож-
ную).  
 R. Beezer [4] пишет: "Не проблема – просто помните об этом!" (Надо не 
забывать еще и про то, что умножение перестановок в Sage также произво-
дится иначе, и либо смириться, либо - пытаться учесть сразу два фактора.) 
 Попробуйте разобраться в следующих двух списках. 
 

sage: S4.cosets(D4,side='right'); 
      S4.cosets(D4,side='left') 
 [ 
 [(),(2,4),(1,2)(3,4),(1,2,3,4),(1,3),(1,3)(2,4), 
  (1,4,3,2),(1,4)(2,3)], 
 [(3,4),(2,3,4),(1,2),(1,2,4),(1,4,3),(1,4,2,3), 
  (1,3,2),(1,3,2,4)], 
 [(2,3),(2,4,3),(1,3,4,2),(1,3,4),(1,2,3), 
  (1,2,4,3),(1,4,2),(1,4)] 
] 
 

[ 
 [(),(2,4),(1,2)(3,4),(1,2,3,4),(1,3),(1,3)(2,4), 
  (1,4,3,2),(1,4)(2,3)], 
 [(3,4),(2,4,3),(1,2),(1,2,3),(1,3,4),(1,3,2,4), 
  (1,4,2),(1,4,2,3)], 
 [(2,3),(2,3,4),(1,2,4,3),(1,2,4),(1,3,2), 
  (1,3,4,2), (1,4,3), (1,4)] 
] 
 Для групп невысокого порядка Sage в состоянии вывести списком все 
их подгруппы (все нормальные подгруппы). В группе ܦସ имеется десять раз-
личных подгрупп, среди них – шесть нормальных. 
 

sage: D4sub=D4.subgroups(); D4sub; len(D4sub) 
 

[ 
 Permutation Group with generators [()],  
 Permutation  Group with generators [(1,3)(2,4)], 

 Permutation Group with generators [(2,4)], 
 Permutation Group with generators [(1,3)], 
 Permutation Group with generators [(1,2)(3,4)],  
 Permutation Group with generators [(1,4)(2,3)], 
 Permutation Group with generators [(2,4),(1,3)(2,4)], 
 Permutation Group with generators [(1,2,3,4), 
  (1,3)(2,4)], 
 Permutation Group with generators [(1,2)(3,4), 
  (1,3)(2,4)],  
 Permutation Group with generators [(2,4),(1,2,3,4), 
  (1,3)(2,4)] 
] 
10 
 

sage: D4subn=D4.normal_subgroups(); D4subn; len(D4subn) 
 

[ 
 Subgroup of (Dihedral group of order 8 as a  
  permutation group) generated by [()], 
 Subgroup of (Dihedral group of order 8 as a 
  permutation group) generated by [(1,3)(2,4)], 
 Subgroup of (Dihedral group of order 8 as a  
  permutation group) generated by [(1,2)(3,4), 
   (1,3)(2,4)],  
 Subgroup of (Dihedral group of order 8 as a  
  permutation group) generated by [(2,4),(1,3)(2,4)], 
 Subgroup of (Dihedral group of order 8 as a  
  permutation group) generated by [(1,2,3,4), 
   (1,3)(2,4)], 
 Subgroup of (Dihedral group of order 8 as a  
  permutation group) generated by [(1,2,3,4), 
   (1,4)(2,3)] 
] 
6 
 Повторим наши вычисления для подгруппы ܦସ применительно к под-
группе четных перестановок ܣସ. Эта подгруппа является нормальной и пото-
му левое и правое фактор-множества в данном случае совпадают. На фактор-
множестве (здесь оно одно) корректно определяется структура группы. Так 
получается фактор-группа ܵସ ∕  ସ. (И мы знаем, что это за группа! Но знаетܣ
ли про это Sage?) 

sage: A4=AlternatingGroup(4); A4; A4.order(); 
      LA4=A4.list(); LA4;  
      A4.is_subgroup(S4), A4.is_cyclic(), 
      A4.is_abelian(), A4.is_normal(S4); 
      A4.gens(); 
      H=S4.quotient(A4); H; H.order(); H.is_ cyclic () 
 

Alternating group of order 4!/2 as a permutation group 
12 
[(),(2,3,4),(2,4,3),(1,2)(3,4),(1,2,3),(1,2,4),(1,3,2), 
 (1,3,4),(1,3)(2,4),(1,4,2),(1,4,3),(1,4)(2,3)] 
(True, False, False, True) 
[(2,3,4),(1,2,3)] 
Permutation Group with generators [(1,2)] 
2 
True 
 Да, знает. Получилась циклическая фактор-группа второго порядка, 
изоморфная ℤ∗ = {1, −1}, но здесь она реализована и описывается как груп-
па ܵଶперестановок степени два. Предлагаем читателям опознать и исследо-
вать еще одну ("именную") подгруппу в ܵସ (в учебниках по теории групп она 
неизменно фигурирует, поскольку служит для демонстрации нетранзитив-
ности отношения "быть нормальной подгруппой"). 
 

sage: V4=PermutationGroup(['(1,2)(3,4)','(1,3)(2,4)']) 
 

Пример 2.4.2. Матричные группы над конечными полями. 
 

 Группа обратимых матриц над конечным полем сама является конеч-
ной и может быть успешно исследована средствами Sage. Рассмотрим об-
щую линейную (general linear) группу 2)ܮܩ, ॲଶ) над полем из двух эле-
ментов. 
 

sage: G=GL(2,2); G; L=G.list(); L; G.order(); 
      G.is_abelian();  
 

General Linear Group of degree 2 over Finite Field of 
 size 2 
( 
[0 1]  [0 1]  [1 0]  [1 0]  [1 1]  [1 1] 
[1 0], [1 1], [0 1], [1 1], [0 1], [1 0] 
) 

6 
False 
 Получилась неабелева группа порядка 6. Доступна таблица умножения, 
которая для групп называется таблицей Кэли. 
 

sage: G.cayley_table(); g=G.gens(); g 
 

*  a b c d e f 
 +------------ 
a| c e a f b d 
b| d f b e a c 
c| a b c d e f 
d| b a d c f e 
e| f d e b c a 
f| e c f a d b 
 

 Известно, что единственной, с точностью до изоморфизма, неабелевой 
группой шестого порядка является группа перестановок ܵଷ. Результат 
,2)ܮܩ ॲଶ) ≅ ܵଷ можно получить средствами Sage, если предварительно за-
дать группу матриц G с помощью порождающих элементов (метод .gens), а 
потом представить заданную генераторами группу как группу перестановок 
(метод .as_permutation_group). 
 

sage: g=G.gens(); g 
 

( 
[1 1]  [0 1] 
[0 1], [1 0] 
) 
 Найдены две матрицы, являющиеся генераторами для G, Теперь мы 
сменим описание группы, она будет задаваться через генераторы; придется 
изменить имя, но мы тут же убедимся в том, что имя H относится к той же 
самой группе 
 

sage: H=MatrixGroup(g[0],g[1]); H; H==G; 
 

Matrix group over Finite Field of size 2 with 2 
 generators  
( 
[1 1]  [0 1] 
[0 1], [1 0] 

) 
True 
 А теперь снова сменим описание: конечную матричную группу пред-
ставим с помощью изоморфной ей группы перестановок.  
 

sage: K=H.as_permutation_group(); K; 
      K==SymmetricGroup(3) 
 

Permutation Group with generators [(2,3),(1,2)] 
True 
 

 И в заключение – толика информации о работе с бесконечными груп-
пами. (Это возможно! Но только, если группа является конечнопорожденной 
(имеет конечное число генераторов). Между генераторами допускается нали-
чие конечного числа соотношений (называемых определяющими). 
 

Пример 2.4.3. Свободные группы конечного ранга. 
 

 Свободная группа ܨ(ܺ) над конечным алфавитом ܺ = ,ଵݔ} … , -௡} опреݔ
деляется как множество редуцированных слов, каждое из которых является 
конечной последовательностью (списком) символов двух типов: букв ݔ௜ и 
формально обратных к ним элементов ݔ௜

ିଵ; редуцированность проявляется в 
том, что два взаимно обратных символа не могут быть соседними. Умноже-
ние слов определяется как их конкатенация (слияние), с последующей ре-
дукцией (сокращением всего, что сокращается); каждое слово оказывается 
произведением своих букв (в степенях ±1); роль нейтрального элемента 
(единицы) играет пустое слово. Мощность алфавита называется рангом сво-
бодной группы. Две свободные группы изоморфны тогда и только тогда, ко-
гда имеют одинаковые ранги. 
 В свободной группе совсем нет определяющих соотношений. Всякая 
группа может быть представлена как фактор-группа свободной группы, что 
на практике сводится к добавлению определяющих соотношений, которым 
должны удовлетворять порождающие элементы (генераторы, они же – буквы 
исходного алфавита). 
 Определим свободную группу ранга 3 с генераторами по умолчанию: 
 

sage: FreeGroup(3) 
 

Free Group on generators {x0, x1, x2} 
 

 А теперь определим свободную группу ранга 3 с заданными именами 
генераторов (ܽ, ܾ, ܿ). Покажем далее, как вводятся и перемножаются элемен-

ты (исходные слова: ݑ = ܾܽܽܽܽିଵܽିଵܾܿ,  ݒ = ܾିଵܿିଵܽିଵܿܿܿܽ; их произведе-
ние: ݒݑ = ܾܽܽܽିଵܽିଵܽିଵܿܿܿܽ). 
 

sage: F.<a,b,c>=FreeGroup(); F;  
      u=F([1,1,2,-1,-1,3,2]); u; 
      v=F([-2,-3,-1,3,3,3,1]); v; u*v 
Free Group on generators {a, b, c} 
a^2*b*a^-2*c*b 
b^-1*c^-1*a^-1*c^3*a 
a^2*b*a^-3*c^3*a 

